/*************************************************************************
FILENAME: FMMR_EJPE_2021_Political_Voice_on_Monetary_Policy.do

This code creates all of the regression tables and figures reported 
  in the paper:

  Ferrara, F. M., Masciandaro, D., Moschella, M., Romelli, D., 2021.
	Political voice on monetary policy: Evidence from the parliamentary hearings of the European Central Bank.
	European Journal of Political Economy, 102143, https://doi.org/10.1016/j.ejpoleco.2021.102143.

  
Before running, please read the README.pdf and 
  overwrite "/SPECIFY/LOCATION/OF/PARENT/DIRECTORY/HERE" in this file with
  the file path being used.
  	

This program is organized into the following sections:

  0. Set library and log file definitions and generate output file templates
  1. Produce summary statistics
  2. Generate Figures 4a and 4b aon the heterogeneity of inflation and unemployment
  5. Run regression analysis 

  
*************************************************************************/


*************************************************************************;
*************************************************************************;
*(0) SET LIBRARY AND LOG FILE DEFINITIONS AND GENERATE OUTPUT FILE TEMPLATES;
*************************************************************************;
*************************************************************************;
*LIBRARY DEFINITIONS;
clear all
clear matrix
clear mata
scalar drop _all
macro drop _all
set mem 2000m
set matsize 10000
set maxvar 10000

*
* Automatically assign directory
dis "current user: `c(username)'"	// Provides the USERNAME of the computer on which Stata is running
if "`c(username)'" == "USERNAME1" {	// Add information on USERNAME1
global dirroot = "/Users/USERNAME/Dropbox/FMMR_EJPE_2021"	// Directory for Mac 
}
else if "`c(username)'" == "USERNAME2" {	// Add information on USERNAME2
global dirroot = "C:\Users\USERNAME2\Dropbox\FMMR_EJPE_2021"	// Directory for Windows
}
*
global dirdta = "${dirroot}/dta"
global dirresults = "${dirroot}/results"
global dirfig = "${dirroot}/figures"
set more off


**********************
* LOCAL DEFINITIONS
*********************

local stats						= 1		// set this local = 1 if you intend to run this part of the cose
local graphs					= 1
local regressions				= 1



*************************************************************************;
*************************************************************************;
*(1) PRODUCE SUMMARY STATISTICS ;
*************************************************************************;
*************************************************************************;

if `stats' == 1 {

* Extract information on the entire database, which also contains the transcripts of the ECB President and Chair of the Econ committee
use "${dirdta}/MonetaryDialogues_Transcripts.dta", clear

* Obtain nr of transcripts
egen dateid=group(date)
sum dateid

* Obtain the number of speeches by the ECB Presidents (2,063)
count if ecb==1

* Obtain the number of speeches by the Chair of ECON (490)
count if Chair==1

use "${dirdta}/Political_Voice_on_Monetary_Policy_Dataset.dta", clear

* Obtain the number of speeches by the MEP part of ECON (1,911)
count if ecb!=1 & Chair!=1

* Obtain number of MEPs (264)
sum name_id

* Obtain number of speeches by men (1,436) and women (475)
tabulate gender, summarize(text_lenght)  nomean nost

* Obtain number of speeches by nationality
tabulate nationality, summarize(nationality) nomean nost

* Obtain number of speeches by political party
tabulate politicalparty, summarize(text) nomean nost

* Obtain number of speeches by by Party Group
* Table 2: Distribution of Monetary Dialogue Speeches by Party Group, 1999-2019
tabulate category, summarize(text) nomean nost

} // end stats
*


*************************************************************************;
*************************************************************************;
*(2) GENERATE FIGURES 4a and 4b ON THE HETEROGENEITY OF INFLATION AND UNEMPLOYMENT;
*************************************************************************;
*************************************************************************;

if `graphs' == 1 {

cd "${dirfigures}"

use "${dirdta}/Political_Voice_on_Monetary_Policy_Dataset", clear
keep if primary_mission!=.
*
* Compute median forecast for Inflation
bysort year: egen Inflation_q_med = median(Inflation_q)
bysort year: egen Inflation_q_max = max(Inflation_q)
bysort year: egen Inflation_q_min = min(Inflation_q)

* Compute median forecast for UnemploymentRate_q
bysort year: egen UnemploymentRate_q_med = median(UnemploymentRate_q)
bysort year: egen UnemploymentRate_q_min = min(UnemploymentRate_q)
bysort year: egen UnemploymentRate_q_max = max(UnemploymentRate_q)

*Graph with min/max values
set scheme s2color
grstyle init
grstyle set plain, horizontal grid
twoway rbar Inflation_q_min Inflation_q_med year, fcolor(gs9) lcolor(gs9) barw(0.6) || rbar Inflation_q_med Inflation_q_max year,  fcolor(gs9) lcolor(gs9) barw(0.6) || scatter Inflation_q_med year, msize(tiny) mcolor(white) || scatter Inflation_q year if Inflation_q<Inflation_q_min & Inflation_q>Inflation_q_max, graphregion(fcolor(gs15)) mcolor(black) msize(tiny) xlabel(1999(2)2019,labsize(small)) ylabel(-6(2)30,labsize(small)) legend(off) xtitle(Year) ytitle(Inflation rate (%)) graphregion(color(white)) saving(inflation, replace)
graph export "${dirfig}/Figure4a.eps", as(eps) preview(on) replace
twoway rbar UnemploymentRate_q_min UnemploymentRate_q_med year, fcolor(gs9) lcolor(gs9) barw(0.6) || rbar UnemploymentRate_q_med UnemploymentRate_q_max year,  fcolor(gs9) lcolor(gs9) barw(0.6) || scatter UnemploymentRate_q_med year, msize(tiny) mcolor(white) || scatter UnemploymentRate_q year if UnemploymentRate_q<UnemploymentRate_q_min & UnemploymentRate_q>UnemploymentRate_q_max, graphregion(fcolor(gs15)) mcolor(black) msize(tiny) xlabel(1999(2)2019,labsize(small)) ylabel(-6(2)30,labsize(small)) legend(off) xtitle(Year) ytitle(Unemployment rate (%)) graphregion(color(white)) saving(unemployment, replace)
graph export "${dirfig}/Figure4b.eps", as(eps) preview(on) replace


} // end graphs
*

*************************************************************************;
*************************************************************************;
*(5) RUN REGRESSION ANALYSIS;
*************************************************************************;
*************************************************************************;

if `regressions' == 1 {


cd "${dirresults}"

use "${dirdta}/Political_Voice_on_Monetary_Policy_Dataset", clear

***********************************************
* Table 4: Macroeconomic Determinants of MEPs' Voice on Primary Mission
***********************************************

** Only Infl.
reg primary_mission lagInflation_q, r
outreg2 using Table4.xls, bdec(4) se replace lab keep(lagInflation_q lagUnemploymentRate_q)
** Infl. + Unemp.
reg primary_mission lagInflation_q lagUnemploymentRate_q, r
outreg2 using Table4.xls, bdec(4) se lab keep(lagInflation_q lagUnemploymentRate_q)
** Add country FEs
reg primary_mission lagInflation_q lagUnemploymentRate_q i.country, r
outreg2 using Table4.xls, bdec(4) se lab keep(lagInflation_q lagUnemploymentRate_q)
** Add time FEs
reg primary_mission lagInflation_q lagUnemploymentRate_q i.country i.date, r
outreg2 using Table4.xls, bdec(4) se lab keep(lagInflation_q lagUnemploymentRate_q)
** Add group FEs, visualises marginal effect of Unemp.
reg primary_mission lagInflation_q lagUnemploymentRate_q i.country i.date i.Group, r
outreg2 using Table4.xls, bdec(4) se lab keep(lagInflation_q lagUnemploymentRate_q)


***********************************************
* Figure 5: Marginal Effect of Unemployment on the Predicted Share of MEPs' Voice on Primary Mission
***********************************************

margins, at(lagUnemploymentRate_q = (4(2)20))
marginsplot, xtitle("Domestic Unemployment") ytitle("Primary Mission")
graph export "${dirfig}/Figure5.eps", as(eps) preview(off) replace


*
**
***********************************************
* Figure 6: MEPs' Voice on Conventional and Unconventional Monetary Policy as a Function on Inflation and Unemployment
***********************************************
**
*
******************************
*** Column 1: Conventional Monetary Policy
******************************
*
**
*** Quarterly data
use "${dirdta}/Political_Voice_on_Monetary_Policy_Dataset", clear
drop lagInflation lagUnemploymentRate
rename lagInflation_q lagInflation
rename lagUnemploymentRate_q lagUnemploymentRate
reg Topic11 lagInflation lagUnemploymentRate i.country i.date i.Group, r
estimates store Conventional
*
**
*** Annual data
use "${dirdta}/Political_Voice_on_Monetary_Policy_Dataset", clear
reg Topic11 lagInflation lagUnemploymentRate i.country i.date i.Group, r
estimates store ConventionalAnnual
*
**
******************************
*** Column 2: Unconventional Monetary Policy
******************************
*
**
*** Quarterly data
use "${dirdta}/Political_Voice_on_Monetary_Policy_Dataset", clear
drop lagInflation lagUnemploymentRate
rename lagInflation_q lagInflation
rename lagUnemploymentRate_q lagUnemploymentRate
reg Topic5 lagInflation lagUnemploymentRate i.country i.date i.Group, r
estimates store Unconventional
*
**
*** Annual data
use "${dirdta}/Political_Voice_on_Monetary_Policy_Dataset", clear
reg Topic5 lagInflation lagUnemploymentRate i.country i.date i.Group, r
estimates store UnconventionalAnnual


* Create Figure 6
set scheme s2color
grstyle init
grstyle set plain, horizontal grid
coefplot (Conventional, label(Reform) msymbol(O)) (ConventionalAnnual, label(Reform) msymbol(Oh)), bylabel(Conventional Mon. Pol.)|| (Unconventional, label(Reform) msymbol(O)) (UnconventionalAnnual, label(Reform) msymbol(Oh)), bylabel(Unconventional Mon. Pol.)||, drop(*country* *date* *Group* _cons) xline(0) ci(90) byopts(xrescale cols(4)) legend(order(2 "Quarterly data" 4 "Annual data")) graphregion(fcolor(white) ifcolor(white) color(white) icolor(white)) bgcolor(white)
graph export "${dirfig}/Figure6.eps", as(eps) preview(on) replace



*
**
***********************************************
* Figure 7: MEPs’ voice on primary mission and subtopics during Non-crisis and Crisis Years
***********************************************
**
*
use "${dirdta}/Political_Voice_on_Monetary_Policy_Dataset", clear
*
******************************
*** Column 1: Primary Mission
******************************
*
reg primary_mission lagInflation_q lagUnemploymentRate_q i.country i.date i.Group if crisis==0, r
estimates store Noncrisis
*
reg primary_mission lagInflation_q lagUnemploymentRate_q i.country i.date i.Group if crisis==1, r
estimates store Crisis
*
**
******************************
*** Column 2: Conventional Monetary Policy
******************************
*
reg Topic11 lagInflation_q lagUnemploymentRate_q i.country i.date i.Group if crisis==0, r
estimates store NoncrisisTopic11
*
reg Topic11 lagInflation_q lagUnemploymentRate_q i.country i.date i.Group if crisis==1, r
estimates store CrisisTopic11
*
**
******************************
*** Column 3: Unconventional Monetary Policy
******************************
*
*
reg Topic5 lagInflation_q lagUnemploymentRate_q i.country i.date i.Group if crisis==0, r
estimates store NoncrisisTopic5
*
reg Topic5 lagInflation_q lagUnemploymentRate_q i.country i.date i.Group if crisis==1, r
estimates store CrisisTopic5

set scheme s2color
grstyle init
grstyle set plain, horizontal grid
coefplot (Noncrisis, label(Reform) msymbol(O)) (Crisis, label(Reform) msymbol(Oh)), bylabel(Primary Mission)|| (NoncrisisTopic11, label(Reform) msymbol(O)) (CrisisTopic11, label(Reform) msymbol(Oh)), bylabel(Conventional M.P.)|| (NoncrisisTopic5, label(Reform) msymbol(O)) (CrisisTopic5, label(Reform) msymbol(Oh)), bylabel(Unconventional M.P.)||, drop(*country* *date* *Group* _cons) xline(0) ci(90) byopts(xrescale cols(4))  legend(order(2 "Non-crisis" 4 "Crisis")) graphregion(fcolor(white) ifcolor(white) color(white) icolor(white)) bgcolor(white)
graph export "${dirfig}/Figure7.eps", as(eps) preview(on) replace


***********************************************
* Figure 8: Robustness tests: Primary mission share with alternative models
***********************************************
**
*
******************************
*** Column 1: 10 Topics
******************************
*
**
*** Quarterly data
use "${dirdta}/Political_Voice_on_Monetary_Policy_Dataset10topics", clear
drop lagInflation lagUnemploymentRate
rename lagInflation_q lagInflation
rename lagUnemploymentRate_q lagUnemploymentRate
reg primary_mission lagInflation lagUnemploymentRate i.country i.date i.Group, r
estimates store Topic10
*
**
*** Annual data
use "${dirdta}/Political_Voice_on_Monetary_Policy_Dataset10topics", clear
replace lagInflation=Inflation
replace lagUnemploymentRate=UnemploymentRate
reg primary_mission lagInflation lagUnemploymentRate i.country i.date i.Group, r
estimates store Topic10Annual
*
**
******************************
*** Column 2: 12 Topics
******************************
*
**
*** Quarterly data
use "${dirdta}/Political_Voice_on_Monetary_Policy_Dataset12topics", clear
drop lagInflation lagUnemploymentRate
rename lagInflation_q lagInflation
rename lagUnemploymentRate_q lagUnemploymentRate
reg primary_mission lagInflation lagUnemploymentRate i.country i.date i.Group, r
estimates store Topic12
*
**
*** Annual data
use "${dirdta}/Political_Voice_on_Monetary_Policy_Dataset12topics", clear
replace lagInflation=Inflation
replace lagUnemploymentRate=UnemploymentRate
reg primary_mission lagInflation lagUnemploymentRate i.country i.date i.Group, r
estimates store Topic12Annual

set scheme s2color
grstyle init
grstyle set plain, horizontal grid
coefplot (Topic10, label(Reform) msymbol(O)) (Topic10Annual, label(Reform) msymbol(Oh)), bylabel(10 Topics)|| (Topic12, label(Reform) msymbol(O)) (Topic12Annual, label(Reform) msymbol(Oh)), bylabel(12 Topics)||, drop(*country* *date* *Group* _cons) xline(0) ci(90) byopts(xrescale cols(4)) legend(order(2 "Quarterly data" 4 "Annual data")) graphregion(fcolor(white) ifcolor(white) color(white) icolor(white)) bgcolor(white)
graph export "${dirfig}/Figure8.eps", as(eps) preview(on) replace


***********************************************
* Table 5: Robustness tests on macroeconomic determinants of MEPs’ voice on primary mission
***********************************************

use "${dirdta}/Political_Voice_on_Monetary_Policy_Dataset", clear

* Generate inflation deviation from 2% target
gen lagInflation_dev_q = lagInflation_q - 2
replace lagInflation_dev_q = -lagInflation_dev_q if lagInflation_dev_q<0

label var lagInflation_dev_q "Inflation Deviation (t-1)"
replace lagNominalGDP_q = lagNominalGDP_q/1000000000000


*
**
*** Table robustness with GDP growth
** Real GDP Growth
reg primary_mission lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q i.country i.date i.Group, r
outreg2 using Table5.xls, bdec(4) se replace lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
** Real GDP Growth + Credit
reg primary_mission lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q ///
lagCredittoPrivateSectortoGDP i.country i.date i.Group, r
outreg2 using Table5.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
** Real GDP Growth + Credit + Long Term Int. Rate 
reg primary_mission lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q ///
lagCredittoPrivateSectortoGDP LongTermInterestRate i.country i.date i.Group, r
outreg2 using Table5.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
** Real GDP Growth + Credit + Long Term Int. Rate + NPLs
reg primary_mission lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q ///
LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans i.country i.date i.Group, r
outreg2 using Table5.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
*
**
*** Using inflation deviation
**
*
** Real GDP
reg primary_mission lagInflation_dev_q lagUnemploymentRate_q lagRealGDPGrowth_q i.country i.date i.Group, r
outreg2 using Table5.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
** Real GDP Growth + Credit
reg primary_mission lagInflation_dev_q lagUnemploymentRate_q lagRealGDPGrowth_q ///
lagCredittoPrivateSectortoGDP i.country i.date i.Group, r
outreg2 using Table5.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
** Real GDP Growth + Credit + Long Term Int. Rate 
reg primary_mission lagInflation_dev_q lagUnemploymentRate_q lagRealGDPGrowth_q ///
lagCredittoPrivateSectortoGDP LongTermInterestRate i.country i.date i.Group, r
outreg2 using Table5.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
** Real GDP Growth + Credit + Long Term Int. Rate + NPLs
reg primary_mission lagInflation_dev_q lagUnemploymentRate_q lagRealGDPGrowth_q ///
LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans i.country i.date i.Group, r
outreg2 using Table5.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)


***********************************************
* Table 6: Robustness Tests: Elections and MEPs’ Voice on Primary Mission
***********************************************

** Only Infl.
reg primary_mission c.lagInflation_q##i.legelec, r
outreg2 using Table6, bdec(4) se replace lab keep(lagInflation_q lagUnemploymentRate_q c.lagInflation_q##i.legelec c.lagUnemploymentRate_q##i.legelec legelec)
** Infl. + Unemp.
reg primary_mission c.lagInflation_q##i.legelec c.lagUnemploymentRate_q##i.legelec, r
outreg2 using Table6.xls, bdec(4) se lab keep(lagInflation_q lagUnemploymentRate_q c.lagInflation_q##i.legelec c.lagUnemploymentRate_q##i.legelec legelec)
** Add country FEs
reg primary_mission c.lagInflation_q##i.legelec c.lagUnemploymentRate_q##i.legelec i.country, r
outreg2 using Table6.xls, bdec(4) se lab keep(lagInflation_q lagUnemploymentRate_q c.lagInflation_q##i.legelec c.lagUnemploymentRate_q##i.legelec legelec)
** Add time FEs
reg primary_mission c.lagInflation_q##i.legelec c.lagUnemploymentRate_q##i.legelec i.country i.date, r
outreg2 using Table6.xls, bdec(4) se lab keep(lagInflation_q lagUnemploymentRate_q c.lagInflation_q##i.legelec c.lagUnemploymentRate_q##i.legelec legelec)
** Add group FEs, visualises marginal effect of Unemp.
reg primary_mission c.lagInflation_q##i.legelec c.lagUnemploymentRate_q##i.legelec i.country i.date i.Group, r
outreg2 using Table6.xls, bdec(4) se lab keep(lagInflation_q lagUnemploymentRate_q c.lagInflation_q##i.legelec c.lagUnemploymentRate_q##i.legelec legelec)


***********************************************
* Table 7: Macroeconomic determinants of MEPs’ voice on secondary and ancillary missions.
***********************************************

** Secondary time FEs
reg secondary_mission lagInflation_q lagUnemploymentRate_q i.country i.date, r
outreg2 using Table7.xls, bdec(4) se lab replace keep(lagInflation_q lagUnemploymentRate_q)
** Secondary group FEs, visualises marginal effect of Unemp.
reg secondary_mission lagInflation_q lagUnemploymentRate_q i.country i.date i.Group, r
outreg2 using Table7.xls, bdec(4) se lab keep(lagInflation_q lagUnemploymentRate_q)
** Ancillary time FEs
reg ancillary_mission lagInflation_q lagUnemploymentRate_q i.country i.date, r
outreg2 using Table7.xls, bdec(4) se lab keep(lagInflation_q lagUnemploymentRate_q)
** Ancillary group FEs, visualises marginal effect of Unemp.
reg ancillary_mission lagInflation_q lagUnemploymentRate_q i.country i.date i.Group, r
outreg2 using Table7.xls, bdec(4) se lab keep(lagInflation_q lagUnemploymentRate_q)


***********************************************
* Table C.1: Robustness tests on macroeconomic determinants of MEPs’ voice on primary mission — generalized linear model estimations
***********************************************

** Real GDP Growth
glm primary_mission lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q i.country i.date i.Group, r link(logit) family(binomial)
outreg2 using AppendixTableC1.xls, bdec(4) se replace lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
** Real GDP Growth + Credit
glm primary_mission lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q ///
lagCredittoPrivateSectortoGDP i.country i.date i.Group, r link(logit) family(binomial)
outreg2 using AppendixTableC1.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
** Real GDP Growth + Credit + Long Term Int. Rate 
glm primary_mission lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q ///
lagCredittoPrivateSectortoGDP LongTermInterestRate i.country i.date i.Group, r link(logit) family(binomial)
outreg2 using AppendixTableC1.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
** Real GDP Growth + Credit + Long Term Int. Rate + NPLs
glm primary_mission lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q ///
LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans i.country i.date i.Group, r link(logit) family(binomial)
outreg2 using AppendixTableC1.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
*
**
*** Using inflation deviation
**
*
** Real GDP
glm primary_mission lagInflation_dev_q lagUnemploymentRate_q lagRealGDPGrowth_q i.country i.date i.Group, r link(logit) family(binomial)
outreg2 using AppendixTableC1.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
** Real GDP Growth + Credit
glm primary_mission lagInflation_dev_q lagUnemploymentRate_q lagRealGDPGrowth_q ///
lagCredittoPrivateSectortoGDP i.country i.date i.Group, r link(logit) family(binomial)
outreg2 using AppendixTableC1.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
** Real GDP Growth + Credit + Long Term Int. Rate 
glm primary_mission lagInflation_dev_q lagUnemploymentRate_q lagRealGDPGrowth_q ///
lagCredittoPrivateSectortoGDP LongTermInterestRate i.country i.date i.Group, r link(logit) family(binomial)
outreg2 using AppendixTableC1.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)
** Real GDP Growth + Credit + Long Term Int. Rate + NPLs
glm primary_mission lagInflation_dev_q lagUnemploymentRate_q lagRealGDPGrowth_q ///
LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans i.country i.date i.Group, r link(logit) family(binomial)
outreg2 using AppendixTableC1.xls, bdec(4) se lab keep(lagInflation_dev_q lagInflation_q lagUnemploymentRate_q lagRealGDPGrowth_q LongTermInterestRate lagCredittoPrivateSectortoGDP lagNonperformingLoans)


} // end regressions
*

